
9 ANHANG: MAKRO, KS 400
#--------------------------------------------------
# MACRO: CART001.MCR
# KEYWORDS: Color Field-measurement of Cartilage (Knorpel)
# ABSTRACT: Field measurement (absolut, mit ROI-Polygon)
# SYSTEM: KS400 - rel 3.0
# OPTIONS: Colour
# IMAGES: RGB-Input (3CCD-Kamera MC3254)
# LUT:
# AUTHORS:  Justin M Burns,  T. Betz
# USER: Dr. Bail, Charité/Virchow-Klinikum Berlin, Unfallchirugie
# --------------------------------------------------------------------------------------------
-----------

# Initialisierung
#--------------------------------------------------
imgdelete "*"
Gclear 0
MSload "default"
update
showwindow "Display",1

# Datenbank Definitionen
# -------------------------------------------------------------------------
# Eingabe, Prüfen und Setzen des Datenpfades
# --------------------------------------------------------------------------
DATAPATH:="D:/daten/cartilage"
while 1
beep 100,1100
beep 100, 990
read DATAPATH,"Geben Sie den Daten-Pfad ein: \n SYNTAX  EINHALTEN !"
    if fexist (DATAPATH) != 2
        beep 300,440
        beep 300,440
        MBok "Der Pfad >> " + DATAPATH + " <<  existiert nicht. Korrigieren Sie Ihre Eingabe."
    else
        break
    endif
endwhile
DBsetpath DATAPATH

# Variablen für Datenbank
# ----------------------------------------------------------------------
# APPEND wird nach der ersten Schleife auf 1 gesetzt!
APPEND=0
# Variable PRO_NR für Projektnummer
PRO_NR=001
# Variable PAT_NR für Patientennummer
PAT_NR=001
# Variable LOKAL für Lokalisation
LOKAL=1
# Variable CUT_NR für Schnittnummer
CUT_NR=001
# Variable DEFECT_AREA für Defektfläche (aufgezogenes Polygon)
DEFECT_AREA=0.0
# Variable CALLUS_AREA für Callusfläche (schwarz gefüllt)
# CALLUS_AREA=0.0
# Variable KNOCHEN_AREA für Knochenfläche (schwarz ohne Füllung)
# KNOCHEN_AREA=0.0
# Variable für DENS_CALLUS für Flächendichte des Calllus
# (= Knochenfläche/Callusfläche * 100%)
DENS_CALLUS=0.0
# Variable KNORPEL für Knorpelfläche (blau)
KNORPEL=0.0
# Variable BINDEGEW  fürBindegewebe (Rot minus Knorpel)
BINDEGEW=0.0
# Variable NONFULL für FIBROUS TISSUE
NONFULL=0.0
RES=0.0
# Eingabe des Datenbanknamens
# --------------------------------------------------------------------
DATABASE="Cart_001"
read DATABASE, "Geben Sie den Datenbanknamen ein!"
# Datenbankstrukturdefinition
DBnew DATABASE,13

DBsetcolumn DATABASE,1,"Projekt_Nr","Int","<none>"
DBsetcolumn DATABASE,2,"Patient_Nr","Int","<none>"
DBsetcolumn DATABASE,3,"Lokalisation","Int","<none>"
DBsetcolumn DATABASE,4,"Schnitt_Nr","Int","<none>"
DBsetcolumn DATABASE,5,"Defektfläche","Float","mm²"
DBsetcolumn DATABASE,6,"Callusfläche","Float","mm²"
DBsetcolumn DATABASE,7,"Knochenfläche","Float","mm²"
DBsetcolumn DATABASE,8,"Callusdichte","Float","%"
DBsetcolumn DATABASE,9,"Knorpel","Float","mm²"
DBsetcolumn DATABASE,10,"Bindegewebe","Float","mm²"
DBsetcolumn DATABASE,11,"Nichtfüllung","Float","mm²"
DBsetcolumn DATABASE,12,"Fib_Tiss","Float","mm²"
DBsetcolumn DATABASE,13,"Resorption","Float","mm²"
datalist DATABASE,0,0
update

# Initialisierung der Meßparameter
# -----------------------------------------------------
MSload "default"
MSsetprop "REGIONFEAT","AREA"
MSsetprop "FIELDFEAT","Flaeche[UNIT^2]=SUM(AREA)"
MSsetprop "DRAWFEAT","DRCONTOUR"
# diese Scalierung paßt zum Objektiv 1.6 und Zoom 0.55
MSsetprop "SCALEX",0.0093633
MSsetprop "SCALEY",0.0093633
MSsetprop "UNIT","millimeter"

# Berechnen des Rectangle 6 x 1,5 mm²
# -------------------------------------------------------------
RECTX=6/0.0093633
RECTY=1.5/0.0093633
# write RECTX
# write RECTY
# Generieren 641 x 160 Pixel !
SIZEX=int(RECTX)
SIZEY=int(RECTY)
# write SIZEX
# write SIZEY

beep 100,660
showwindow "Messages",1
write "@"
write "Stellen Sie das Mikroskop wie folgt ein!"
write "Objektiv 1,6x, TV-Zoom: 0,55x"
write "Tubus auf 50/50 stellen!"
write "Beleuchtung auf 7 V , Graufilter ON"
write "Leuchtfeld- und Aperturblende voll auf!!!"
write "Neutralfilter 0.4 + 16 im Durchlichtstrahlengang rein"
write "Kamera MC 3254 auf Normalwerte : Gain=0 ...."
beep 100,1000
MBok "Haben Sie alles eingestellt ?"
showwindow "Messages",0
# --------------------------------------------------------------------------------------------
----
# Meßschleife
while 1
    # das Macro ist so aufgebaut, daß alle Schnitte in einer
    # Datenbank gemessen werden!
    # deshalb werden erst hier die allgemeinen
    # Parameter abgefragt
    # ----------------------------------------------------------------------------------
    # neue Zeile in Datenbank nicht vergessen!
    DBaddline DATABASE
    #
    read PRO_NR,"Geben Sie die Projekt-Nummer ein: \n SYNTAX  EINHALTEN !"
    DBsetvalue DATABASE,"Projekt_Nr",PRO_NR
    read PAT_NR,"Geben Sie die Patienten-Nummer ein: \n SYNTAX  EINHALTEN !"
    DBsetvalue DATABASE,"Patient_Nr",PAT_NR
    read LOKAL,"Geben Sie die Lokalisation ein: \n SYNTAX  EINHALTEN !"
    DBsetvalue DATABASE,"Lokalisation",LOKAL
    read CUT_NR,"Geben Sie die Schnitt-Nummer ein: \n SYNTAX  EINHALTEN !"
    DBsetvalue DATABASE,"Schnitt_Nr",CUT_NR
    #

while 1
    #Bildaufnahme
    imgdelete "*"

    Gclear 0
    update
    # Anlegen eines Hilfsbildes für Wincopy
    imgnew "FRAME",758,572,1,"Colour"
    beep 100,880
    beep 150,660
    pause "Stellen Sie das nächste Bildfeld ein!",2000
    # pause "Laden des Beispielbildes von der Harddisk!",2000

    ! tvinput "INPUT"
    # !imgload "k:\bail\cartilage\cart_02.tif","INPUT"
    imgdisplay "INPUT"
    # Shading-Korrektur    (bei unbedingter  Notwendigkeit)
    lowpass "INPUT","SHAD_REF",49,5
    shadcorr "INPUT","SHAD_REF","INP_GOOD",2,0
    # Enhancement
    #--------------------------------------------------
    normalize "INPUT","ENH_NORM",5
    # Kopieren des Bildes in das FRAME-Hilfsbild
    wincopy "ENH_NORM","FRAME",6,4,758,572,0,0
    while 1
        Gclear 0
            imgdelete "RECT"
            imgdisplay "FRAME"
            # Maskierung erzeugen für Defektfläche
            # --------------------------------------------------------------
            beep 100,880
            beep 150,660
            pause "Verschieben Sie das Rechteck und \n ziehen Sie den oberen Rand bei Bedarf
auf!",0
            SIZEX=int(RECTX)
            SIZEY=int(RECTY)
            STARTX=50
            STARTY=250
            ! Grectg STARTX,STARTY,SIZEX,SIZEY,14,0
            Gmerge "RECT",255
    MByesno "Do you need to correct the rectangle image?"
        if _STATUS!=1:break
            APPEND=1
    endwhile

    MByesno "Do you need to correct the tv image?"
        if _STATUS!=1:break
            APPEND=1
    endwhile

    dislevrgb "RECT",21,0,0,255,255,255,255,0,0,10,"RGB"
    binfill 21,29

    # Erzeugen des Meßbildes (Maskierung)
    # ----------------------------------------------------------------
Gclear 0
    imgnew "MEAS_SM",SIZEX,SIZEY,1,"Colour"
    wincopy "FRAME","MEAS_SM",STARTX,STARTY,SIZEX,SIZEY,0,0
    imgdisplay "MEAS_SM"

    MByesno "Do you need to correct the defect area?"
if _STATUS == 1
    while 1
        Gclear 0
        imgdelete 1
        imgdelete "MASK1"
        imgdelete "MASKFILL"
        imgdelete 20
        pause "Outline the defect area with a polygon, pensize 2, and light green.  Do not go
off the screen",0
        imgcopy "MEAS_SM",1
        imgedit 1
        Gmerge "MASK1",255
        binfill "MASK1","MASKFILL"
        imgcopy "MASKFILL",20
        imgdisplay 20

        MByesno "Do you need to correct this image?"
        if _STATUS!=1:break
            APPEND=1
        endwhile
else

    dislevrgb "MEAS_SM","MASK1",0,0,255,255,255,255,0,0,10,"RGB"
    binnot "MASK1","MASK1"
    binfill "MASK1","MASKFILL"
    imgcopy "MASKFILL",20

endif

    # Messen der Defektfläche
wait 1500
update
    MSmeasmask "MASKFILL","FRAME","MASKFILL",0,1,10
    DBgetvalue "MASKFILL","AREA",MASKAREA
    # write MASKAREA
    DBsetvalue DATABASE,"Defektfläche",MASKAREA
wait 1000

    # Segmentieren der Knochenfläche
    # --------------------------------------------------------------
    while 1
        imgdisplay "MEAS_SM"
        Gclear 0
        imgdelete "MEASKN1"
        imgdelete "MEASKN2"
        showwindow "Messages",1
        write "@"
        write "You are now measuring the"
        write "bone area"
        beep 100,880
        beep 150,660
        wait 1500
        ! dislevrgb "MEAS_SM","MEASKN1",1,0,0,255,11,101,0,123,10,"HLS"
        binscrap "MEASKN1","MEASKN2",0,30,0
        binnot "MEASKN2","MEASKN2"
        binfill "MEASKN2","MEASKN2"
        binnot "MEASKN2","MEASKN2"
        imgdisplay "MEASKN2"
        MByesno "This is the measurement image. Does it need to be corrected?"
                if _STATUS!=1:break
                    APPEND=1
    endwhile
    # Messen der Knochenfläche (schwarz)
    # --------------------------------------------------------------------------------
    MSmeasmask "MEASKN2","FRAME","KNOCHEN",0,2,10
    DBgetvalue "KNOCHEN","Flaeche",KNAREA
    # write KNAREA
    DBsetvalue DATABASE,"Knochenfläche",KNAREA
    # Ermittlung der Callusfläche durch interakives Schließen
    # ----------------------------------------------------------------------------------------
    while 1
        imgdelete "MEASKN3"
        imgdelete "MEASKN4"
        imgdelete "MEASKN5"
        imgdelete "MEASKN6"
    while 1
        Gclear 0
        imgdelete "MEASKN21"
        imgcopy "MEAS_SM",1
        imgdisplay 1
        beep 100,880
        beep 150,660
        pause "Schließen Sie die Knochenfläche mit dem Stift \n in Stärke 2 und mit der
hellgrünen Farbe!",0
        imgedit 1
        Gmerge "MEASKN21",255
        imgdisplay "MEAS_SM"
            MByesno "Do you need to correct this image?"
            if _STATUS!=1:break
                APPEND=1
    endwhile
Gclear 0

binor "MEASKN21","MEASKN2","MEASKN21"
binclose "MEASKN21","MEASKN21",7,1
binfill "MEASKN21","MEASKN3"
Gmerge "MEASKN3",255
Gclear 0
binnot "MEASKN3","MEASKN4"
binscrap "MEASKN4","MEASKN5",0,10000,0
binnot "MEASKN5","MEASKN6"
imgdisplay "MEASKN6"
wait 1000
            MByesno "This is the image to be measured.  Does it need to be corrected?"
            if _STATUS!=1:break
                APPEND=1
    endwhile

MSmeasmask "MEASKN6","FRAME","CALLUS",0,2,10
DBgetvalue "CALLUS","Flaeche",CALAREA
DBsetvalue DATABASE,"Callusfläche",CALAREA

DENS_CALLUS=KNAREA/CALAREA*100
DBsetvalue DATABASE,"Callusdichte",DENS_CALLUS
Gclear 0

#  Measuring the Nichtfüllung
    while 1
        Gclear 0
        imgdelete "REST"
        imgdisplay "MEAS_SM"
        beep 100,660
        showwindow "Messages",1
        write "@"
        write "You are now measuring the"
        write "Nichtfüllung"
        wait 1500
        ! dislevrgb "MEAS_SM","REST",1,0,0,108,77,255,0,255,10,"HLS"
        binscrap "REST","REST",0,2000,0
        binfill "REST","REST"
        binnot "REST","REST"
        binscrap "REST","REST",0,200,0
        binnot "REST","REST"
        binand "REST",20,"REST"
        imgdisplay "REST"
        MByesno "This is the measurement image. Does it need to be corrected?"
            if _STATUS!=1:break
                APPEND=1
    endwhile

MSdrawmask "REST","FRAME"
Gmerge 3,255
imgdisplay "MEAS_SM"
MSmeasmask "REST","FRAME","REST",0,2,10
DBgetvalue "REST","Flaeche",REST
DBsetvalue DATABASE,"Nichtfüllung",REST
Gclear 0

# Measuring the Bindegewebe
binor "REST","MEASKN6","BG"
binnot "BG","BG"
MSdrawmask "BG","FRAME"
Gmerge 4,255
imgdisplay "MEAS_SM"
MSmeasmask "BG","FRAME","BG",0,2,10
DBgetvalue "BG","Flaeche",BG
DBsetvalue DATABASE,"Bindegewebe",BG
Gclear 0
update

#  Bone Resorption
imgdisplay "ENH_NORM"
MByesno "Do you need to measure the bone resorption?"
if _STATUS == 1
    while 1
        Gclear 0
        imgdelete "POLY1"
        imgdelete "21"
        beep 100,660
        showwindow "Messages",1

        write "@"
        write "Select the polygon, 2, and light green"
        write "Position the polygon so that the bone resorption"
        write "area is enclosed with in."
        wait 1500
        imgdisplay "ENH_NORM"
        imgcopy "ENH_NORM",21
        imgedit 21
        Gmerge "POLY1",255
        imgdisplay "ENH_NORM"
    MByesno "Do you need to correct this image?"
        if _STATUS!=1:break
            APPEND=1
    endwhile
        Gclear 0
        binfill "POLY1",22
        dislevrgb "ENH_NORM",23,1,0,68,52,3,255,0,171,10,"HLS"
        binscrap 23,24,0,70,0
        binfill 24,24
        binnot 24,24
        binmask 22,24,26
        subtract 26,29,27,2

else
        imgdisplay "ENH_NORM"
        Gmerge "POLY2",255
        Gclear 0
        binfill "POLY2",27
endif

imgdisplay "ENH_NORM"
MSmeasmask 27,"FRAMEB",27,0,2,10
DBgetvalue 27,"Flaeche",RES
DBsetvalue DATABASE,"Resorption",RES
update
wait 1000

# Green
beep 100,880
beep 200,990
pause "Insert Safranin-O/ Light green image",0
Gclear 0
    ! tvinput "INPUT"
imgcopy "INPUT","INPUTB"
    # !imgload "k:\bail\cartilage\cart_02B.tif","INPUTB"
    imgdisplay "INPUTB"
    # Enhancement
    #--------------------------------------------------
    normalize "INPUTB","ENH_NORB",5
    # Anlegen eines Hilfsbildes für Wincopy
    imgnew "FRAMEB",758,572,1,"Colour"
    # Kopieren des Bildes in das FRAME-Hilfsbild
    wincopy "ENH_NORB","FRAMEB",6,4,758,572,0,0
Gclear 0

! dislevrgb "FRAMEB","CART",1,0,184,255,61,189,65,255,10,"HLS"
binscrap "CART",11,0,20,0
binfill 11,11

    while 1
        Gclear 0
        imgdelete "CARTEND"
        imgdelete "POLY"
        imgdelete "MASK200"
        beep 100,660
        showwindow "Messages",1
        write "@"
        write "Now you must outline the new cartilage!"
        write "Select the polygon, 2, and light green"
        write "Position the polygon so that the"
        write " new cartilage area is enclosed with in."
        wait 1500
        imgcopy "FRAMEB",1
        imgedit 1

        Gmerge "POLY",255
        MSdrawmask "POLY","FRAMEB"
        imgdisplay "FRAMEB"
    MByesno "Do you need to correct this image?"
        if _STATUS!=1:break
            APPEND=1
    endwhile
            Gclear 0
            binfill "POLY","MASK200"
            binand "MASK200",11,"CARTEND"
            imgdisplay "FRAMEB"
            MSmeasmask "CARTEND","FRAMEB",11,0,2,10
            DBgetvalue 11,"Flaeche",CART
            DBsetvalue DATABASE,"Knorpel",CART
            Gclear 0
            update
#  Fibrous Tissue Measurement

DBgetvalue DATABASE,"Knorpel",CART
DBgetvalue DATABASE,"Bindegewebe",BG
BINDEGEW=BG
KNORPEL=CART
NONFULL=BINDEGEW-KNORPEL
wait 2000
DBsetvalue DATABASE,"Fib_Tiss",NONFULL
update
wait 2000
# Erzeugen des Meßbildes für rotes Bindegewebe
    update
    MByesno "Wollen Sie ein weiteres Bildfeld vermessen?"
    if _STATUS!=1:break
    APPEND=1
endwhile
beep 100,880
beep 200,990
showwindow "Messages",1
write "@"
write "Ende des Macros"
write "Exportieren Sie die Daten!"
wait 2000
stop


